
News  
Spike2 and Signal Training Courses 

All available places for Spike2 sessions at our 2007 U.K. training days have now been 
fully booked. We are still able to take limited bookings for Signal and workshop sessions 
via the training section of our website. The courses will take place on the 12th & 13th of 
April at St. John’s Innovation Centre, Cambridge and we look forward to seeing all 
registered attendees on these dates. If you would like to be added to the mailing list to 
receive information about future CED training days, please contact us at 
sales@ced.co.uk.  

 

Meetings and events 
German Neuroscience Society 2007                                                                     
Gottingen, Germany 
March 28th – April 1st 
British Neuroscience Association 2007                                                                    
Booth No. 4  
Harrogate International Centre, Harrogate, UK 
April 1st – 4th 2007 

Experimental Biology 2007                                                                                         
Booth No. 1831 
Washington, DC, USA 
April 28th – May 2nd 2007 

8th Meeting of the French Neuroscience Society 2007                                                                
Corum-Esplanade Charles de Gaulle 
Montpellier, France 
May 22nd – 25th 2007 

 

  Latest versions of Spike2 and Signal 
Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.  
 

 Spike2 downloads Signal downloads 
Spike2 version 6.03  Signal version 3.09 

 Spike2 version 5.17  Signal version 2.16 

 Spike2 demo  Signal demo 

A full list of the new features and changes in the software versions is available from 
the website. 

Spike2   

 

Q. Is there any way to position vertical and horizontal cursors at specific locations by 
typing in an x or y-axis value, rather than trying to place them at exact positions 
by clicking and dragging? 

A. In previous versions of Spike2 you had to use the associated script commands to 
enter specific positions for cursors. You can now position vertical and horizontal 
cursors interactively, either from the Cursor menu or by right clicking on a cursor 
in a data, result or XY-view. Cursor positions can be entered as x or y-axis 
values or expressions such as Cursor(0)+1.5. 

 
Interactive dialog for cursor position 

 

There are also new interactive commands for cursors which allow you to set 
user-defined labels. There are additional options to display channel y-values as 
part of the label, including pre-set instructions available from the dialog which 
allows the user to ‘build’ labels for cursors based on number, position and Y-
values from multiple channels.  

 
 

An example of user-defined labels showing channel values 

  

 

Q. I would like to be able to analyse the duration of burst events from my single-
channel data in Signal. Is there any way to specify a burst interval parameter so 
that only open/closed times that fall within this interval are included? 

A. From the Open/Closed times option in the Analysis menu you can generate an 
Event burst histogram of open/closed times from an idealized trace channel. You 
can specify which events to include and exclude from the histogram along with 
the maximum duration of a burst (which also sets the histogram width) and the 
critical interval (the time range that an event duration must fall within to be 
included in the result).  

Burst duration histogram and settings dialog 

 

Scripts: Spike2  
Q. I was wondering if there was any way to apply on-line high pass filtering to a 

waveform channel in Spike2 using the built-in digital filters. 

A. The attached script, Online Filter.s2s, is an example of applying IIR filters 
to a waveform channel during sampling. The script opens a new data file based 
on the current sampling configuration and then generates a dialog from which the 
user can select the type of filter to apply, and the source waveform channel to 
use. Once sampling is started the script uses an idle routine to filter the source 
data and produce a new memory channel holding the result. This script requires 
the latest version of Spike2, 6.03. It could also be modified to include FIR digital 
filtering commands. Information on the characteristics of IIR and FIR filters can 
be found in the Spike2 on-line help.  

 

Scripts: Signal  

Q. I record intracellular spike data using Signal and would like some way to analyse 
multiple spike occurrences in a single frame. I have tried using the active cursors 
but it seems that you can only use these to search once per frame. There doesn’t 
seem to be any way of stepping between multiple spikes in a frame using the 
same cursors   

A. The attached script, SpikeAnalysis.sgs, will automatically run through a 
Signal data file finding and taking measurements from each spike. The user sets 
a level threshold using a horizontal cursor and the script then finds each spike 
that exceeds this level. You can check individual spikes for correct detection of 
features and run through all spikes to take measurements using the associated 
toolbar buttons. A text file is then created holding the measurement results and 
time of occurrence for each spike. This script requires Signal version 3. 

opening the embedded 
scripts in this newsletter 
please let us know. 

Back to top 
Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 

'$Online Filter.s2s|Script for applying IIR filters to a waveform channel on-line.
'User selects the type of filter to apply and nominates the source waveform channel.
'The script then writes the filtered data to a new memory channel in the file during sampling

const lo:= 0.0002;		'The minimum low pass setting for filters (as a fraction of the sample rate)
const hi:= 0.4998;		'The maximum high pass setting for filters (as a fraction of the sample rate)

var data%;								'Handle of new data file (Spike2 sets it to 0)
var sTime;								'The last time we looked at the idle routine
var dataFile%,chan%;			
var filtName%,filtType%;		'Variables returned from dialog holding filter name to use and type
var order%,notch%,reson%;		'Variables holding filter settings
var rate,cFreq1,cFreq2,eFreq1,eFreq2,ripple,freq,nFreq,qFact;	'variables holding filter settings
var int;
var nyquist,minVal;
var numPoints%;
var ok%;
var v%,d%,f%,c%;		'variables to hold dialog field return values
var filtCh%;
var type%;
var useFilt%;
var filtList$[4];					'Array to hold filter names
var typeList$[4];                   'Array to hold filter types
var gFloat%[20];              'global for floating window states

if App(-1)<601 then Message("This script needs at least Spike2 version 6.01 to run.\nRegistered version 6 users can download the latest\nversion from our website"); halt endif;	'Checks that the version of Spike2 is suitable


filtList$[0]:="Butterworth";        'List of filters
filtList$[1]:="Bessel";
filtList$[2]:="Chebyshev type 1";
filtList$[3]:="Chebyshev type 2";

typeList$[0]:="Low pass";           'List of filter types
typeList$[1]:="High pass";
typeList$[2]:="Band pass";
typeList$[3]:="Band stop";

HideAll();                       'Hide all unwanted bars
ToolbarVisible(1);				 'Make toolbar visible always
New%();							 'Set up new sampling window
DoToolbar();					 'Do the toolbar
RestoreAll();                    'Restore all bars
Halt;

func New%()							'New sampling window
View(LogHandle());					'Make log view the current view
EditSelectAll();					'Select all text in log view
EditClear();						'Delete it
Window(0,80,100,100);				'Display it at the bottom of the screen
WindowVisible(1);					'Make it visible

if data%>0 then						'If there is already a data view open...
  View(data%);						'...close it
  FileClose();                      'No point setting data% to 0...
endif;

data%:=FileNew(0,1);				'Open a new data file for sampling
if data%<0 then Message("Unable to open new data file");Halt() endif;
DrawMode(-1,2);						'Set event draw mode to lines
Window(0,0,100,80);					'Make data window in top bit of screen
XRange(0,10);

FrontView(LogHandle());				'Bring the Log view to the front
FrontView(data%);					'Bring the data view to the front

IIRSettings%();
useFilt%:=IIRMake%();
filtCh%:= MemChan(0,chan%);
ChanShow(filtCh%);

ToolbarEnable(3,0);					'Disable "Sample stop" button
ToolbarEnable(2,1);					'Disable "Sample stop" button
ToolbarText("Press SAMPLE START to commence sampling");
return 1;
end;


Func IIRSettings%()
repeat		            'If settings are not compatible, the dialog repeats
	DlgCreate("On-line IIR Filter settings");
	DlgGroup("Select Filter Type",1,1,47,4);
	DlgList(1,"",filtList$[],4,4,2);
	DlgList(2,"",typeList$[],4,28,2);
	DlgGroup("Source data",1,5.5,47,4);
	DlgGroup("Filter Characteristics",1,10,47,6);
	DlgCheck(3,"Notch",3,3);
	DlgCheck(4,"Resonator",3,4);
	DlgInteger(5,"Set Filter Order|For Bandpass and Bandstop filters, the Order used is twice this setting",1,10,22,11.2,1);
    DlgChan(6,"Data channel",1+512,20,6.5);
    DlgReal(7,"Edge Frequency 1 (Hz)",0.1,250000,22,12.4,0.1);
	DlgReal(8,"Edge Frequency 2 (Hz)",0.1,250000,22,13.6,0.1);
	DlgReal(9,"Ripple (dB)|This is the desired pass band ripple for type 1, or the cut in the stop band for type 2",1,100,22,14.8,1);
	DlgGroup("Notch and Resonator settings",1,16.5,47,3.5);
	DlgReal(10,"Notch Freq. (Hz)",0.1,250000,22,17.7,0.1);
	DlgReal(11,"q-Factor|The higher the q value, the narrower the notch. If unsure, try 100 as a starting value",10,1000,22,18.9,1);
	DlgAllow(1023,0,DlgSettings%);
	
		
	ok%:=DlgShow(filtName%,filtType%,notch%,reson%,order%,chan%,eFreq1,eFreq2,ripple,Freq,qFact);
	

	if ok%=0 then RestoreAll();halt;endif;		'If user presses cancel, quit script

	if ok%>0 then 				'If user presses OK
            rate:= 1/BinSize(chan%);		'Get sample rate of selected channel
			
		
		if d%=1 or f%=1 then	 'The following statements check for enabled fields, and only calculates frequencies for those
			nFreq:= freq/rate;	 'All unused fields have values set to lo as they are not used
			cFreq1:=lo;
			cFreq2:=lo;
		else;
		
		if d%=0 and f%=0 then
		cFreq1:= eFreq1/rate;	'Calculate used frequencies as a fraction of sample rate to check compatibility
		cFreq2:=lo;
			if v%=2 or v%=3 then
				cFreq2:= eFreq2/rate;	
				else
				cFreq2:=lo;
			endif;
		nFreq:=lo;
		endif;
		endif;
        
        nyquist:= rate/2;
        minVal:= rate/5000;
        
        if cFreq1 > hi or cFreq1 < lo or cFreq2 > hi or cFreq2 < lo or nFreq > hi or nFreq < lo then		'If outside the acceptable range
            if notch% or reson% = 1 then
                Message("The Notch frequency must be less than %01.2f (1/2 the sample rate)\nand greater than %f (1/5000 the sample rate) for the selected data!",nyquist,minVal);	'warn user
            else
                Message("The Edge frequencies must be less than %01.2f (1/2 the sample rate)\nand greater than %f (1/5000 the sample rate) for the selected data!",nyquist,minVal);	'warn user
            endif;    
           ok%:=0;		'Reset ok% value so that dialog repeats
        endif;
                   
	endif;	
until ok%>0;		'until we have acceptable settings to proceed
return ok%;
end;


Func DlgSettings%(item%)'This function monitors user changes to the dialog so that only relevant fields can be selected

v%:= DlgValue(2);		'Get type of filter
d%:= DlgValue(3);		'Check if Notch selected
f%:= DlgValue(4);		'Or Resonator...
c%:= DlgValue(1);		'Get filter name

DlgEnable(0,8,9);

docase
	case d%=1 then			    'If Notch filter selected
		DlgValue(4,0);		    'Clear resonator
		DlgValue(3,1);		    'Set Notch
		DlgEnable(0,1,2,5,7,8);	'Set dialog to initial state
		DlgEnable(1,10,11);

	case f%=1 then			    'If Resonator selected
		DlgValue(3,0);		    'Clear Notch
		DlgValue(4,1);		    'Set Resonator
		DlgEnable(0,1,2,5,7,8);	'Set dialog to initial state
		DlgEnable(1,10,11);

	case d%=0 and f%=0 then		'If neither selected, enable all standard fields
		DlgEnable(1,1,2,5,7);	'Enable Filter name field
		DlgEnable(0,10,11);		'Disable Notch.. & Resonator settings
		
		if v%=2 or v%=3  then	'If Bandpass or Bandstop filter selected..
			DlgEnable(1,8);		'..Enable Edge frequency 2 field
		endif;
		if c%=2 or c%=3 then	'If either Chebyshev filter selected..
			DlgEnable(1,9);		'..Enable ripple field
	 	endif;
endcase;
return 1;
end;



Func IIRMake%()
var filtMessage$;
docase	
	case notch%=1 then	'If Notch selected
		IIRNotch(0,nFreq,qFact);		'Generate Notch filter
        filtMessage$:=Print$("IIR Notch filter with a frequency of %01.2fHz and q-factor of %d",freq,qFact);
        type%:=1;
        
	case reson%=1 then	'If Resonator selected
		IIRReson(0,nFreq,qFact);		'Generate Resonator filter
        filtMessage$:= Print$("IIR Resonator filter with a frequency of %01.2fHz and q-factor of %d",freq,qFact);
        type%:=2;
             
	case filtType%=0 then	'If Lowpass selected
		if filtName%>=2 then		'If Chebyshev type filter..
			IIRLp(0,cFreq1,order%,filtName%,ripple);	'..generate Lowpass filter including ripple setting
			filtMessage$:=Print$("%s Lowpass filter with order of %d, edge frequency of %01.2fHz and ripple of %ddB",filtList$[filtName%],order%,eFreq1,ripple);
            
        else
			IIRLp(0,cFreq1,order%,filtName%);	'Otherwise generate Lowpass filter
			filtMessage$:=Print$("%s Lowpass filter with order of %d and edge frequency of %01.2fHz",filtList$[filtName%],order%,eFreq1); 
            
        endif;
        type%:=3;
             
	case filtType%=1 then	'If Highpass selected
		if filtName%>=2 then		'If Chebyshev type filter..
			IIRHp(0,cFreq1,order%,filtName%,ripple);	'..generate Highpass filter including ripple setting
			filtMessage$:=Print$("%s Highpass filter with order of %d, edge frequency of %01.2fHz and ripple of %ddB",filtList$[filtName%],order%,eFreq1,ripple);
            
        else
			IIRHp(0,cFreq1,order%,filtName%);	'Otherwise generate Highpass filter
			filtMessage$:=Print$("%s Highpass filter with order of %d and edge frequency of %01.2fHz",filtList$[filtName%],order%,eFreq1);
            
        endif;
        type%:=4;
          

	case filtType%=2 then	'If Bandpass selected
		if filtName%>=2 then		'If Chebyshev type filter..
			IIRBp(0,cFreq1,cFreq2,order%,filtName%,ripple);	'..generate Bandpass filter including ripple setting
			filtMessage$:=Print$("%s Bandpass filter with order of %d, edge frequencies of %01.2fHz and %01.2fHz, and ripple of %ddB",filtList$[filtName%],order%,eFreq1,eFreq2,ripple);
            
        else
			IIRBp(0,cFreq1,cFreq2,order%,filtName%);	'Otherwise generate Bandpass filter
			filtMessage$:=Print$("%s Bandpass filter with order of %d and edge frequencies of %01.2fHz and %01.2fHz",filtList$[filtName%],order%,eFreq1,eFreq2); 
           
        endif;
        type%:=5;
    

	case filtType%=3 then	'If Bandstop selected
		if filtName%>=2 then		'If Chebyshev type filter.. 
			IIRBs(0,cFreq1,cFreq2,order%,filtName%,ripple);	'..generate Bandstop filter including ripple setting
			filtMessage$:=Print$("%s Bandstop filter with order of %d, edge frequencies of %01.2fHz and %01.2fHz, and ripple of %ddB",filtList$[filtName%],order%,eFreq1,eFreq2,ripple);
            
        else
			IIRBs(0,cFreq1,cFreq2,order%,filtName%);	'Otherwise generate Bandstop filter
			filtMessage$:=Print$("%s Bandstop filter with order of %d and edge frequencies of %01.2fHz and %01.2fHz",filtList$[filtName%],order%,eFreq1,eFreq2);
            
        endif;
        type%:=6;
endcase;
return type%;
end;

proc DoToolbar()
ToolbarSet(1,"&Quit",Quit%);		'Set up toolbar buttons
ToolbarSet(2,"&Sample start", Start%);
ToolbarSet(3,"&Sample stop", Stop%);
ToolbarSet(4,"&New file", New%);
ToolbarEnable(3,0);					'Disable "Sample stop" button
Toolbar("Press SAMPLE START to commence sampling", 1023);'Wait here until quit
end;

func Quit%()							'If "Quit" is pressed
SampleStop();							'Stop sampling
return 0;								'leave toolbar
end;

func Start%()							'If "Start" is pressed
ToolbarSet(0,"",Idle%);				'Call Idle%() whenever there is free time
sTime := 0;                      'set start of anaylsis time
SampleStart();							'Start sampling
ToolbarEnable(4,0);					'Disable "New file" button
ToolbarEnable(3,1);					'Enable "Sample stop" button
ToolbarEnable(2,0);					'Disable "Sample start" button
ToolbarEnable(1,0);					'Disable "Quit" button
ToolbarText("Press SAMPLE STOP to stop sampling");
return 1;								'Stay with toolbar
end;

func Stop%()							'If "Stop" is pressed
SampleStop();							'Stop sampling
if (data%>0) and                 'if no data in file, stop will close it
   (ViewKind(data%) <> 0) then data% := 0 endif;
ToolbarEnable(4,1);					'Enable "New file" button
ToolbarEnable(3,0);					'Disable "Sample stop" button
ToolbarEnable(1,1);					'Enable "Quit" button
ToolbarClear(0);
ToolbarText("Press FILE NEW to capture more data");
return 1;								'Stay in toolbar
end;

func Idle%()							'Idle routine runs when system has time
var eTime;
var pts%;
var data[rate];

if ViewKind(data%)=0 then        'This checks that the data view exists...
   View(data%);                  '...as this gives an error if it doesn't
   eTime:=Maxtime();

    if (eTime > sTime) then
      pts%:=ChanData(chan%,data[],(sTime-BinSize(chan%)),eTime);
        if pts% > 0 then
            Apply%(useFilt%,data[:pts%]);',pts%);
            ChanWriteWave(filtCh%,data[:pts%],(sTime-BinSize(chan%)));
        endif;
      sTime:=eTime;
      endif;
   endif;
return 1;								'Stay in toolbar
end;



Func Apply%(which%,data[]);
docase
    case which%=1 then
        IIRNotch(data[],2);
    case which%=2 then
        IIRReson(data[],2);
    case which%=3 then
        IIRLp(data[],2);
    case which%=4 then
        IIRHp(data[],2);
    case which%=5 then
        IIRBp(data[],2);
    case which%=6 then
        IIRBs(data[],2);
endcase;    
return 1;
end;

proc HideAll()
var i%;
gFloat%[0] := App(-2);        'number of windows
for i% := 1 to gFloat%[0] do  'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do  'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end




CED
Online Filter.s2s


'$SpikeAnalysis|Script to automatically iterate through and measure multiple spike
'occurences in Signal data frames.  

var count%;					'Count of spikes in frame
var current;				'Current measured spike
var gotSpikeFlag% := 0;		'Flag f a spike is found
var frameCounter%;			'Keep count of frames
var data%;					
var chan% := 1;				'Channel to analyse
var spikeTimes[1000];		'Array to hold actual times of spikes
var printCount%;
var pTime,pAmp;				'Time and level of peak for spike					
var sTime,sAmp;				'Time and level of spike start	
var amp;					'Peak amplitude (pAmp-sAmp)
var RiseAmp10,RiseTime10;	'Level and time at 10% rise
var riseAmp90,RiseTime90;	'     "        "   90%  "				
var RiseAmp50,RiseTime50,FallTime50;	'Level and times for 50% rise and fall
var rTime;					'Time of steepest rising point
var hWidth;					'Half width time
var AHPTime,AHPAmp;			'Time and level of afterhyperpolarisation peak
var maxdata,minData,pDAmp;	'range of data in initial frame. Peak detection amplitude set at range/20
var res%;					'Text file to hold detailed results
var test;					'Check to see if spikes present in frame
var check%;					'Flag to either inspect individual spikes or run through whole file				
var all%;					'Additional variable related to check%
var got%;					'Returned indicating search errors
var error%[7];				'Array of flags indication search errors
var numErrors%;				'Number of detected errors
var i%,err%;
var pos$[7];				
var gFloat%[20];

pos$[0] := "Peak";			
pos$[1] := "Threshold";
pos$[2] := "10% rise";
pos$[3] := "90% rise";
pos$[4] := "50% rise";
pos$[5] := "50% fall";
pos$[6] := "AHP";
var ok%;					'Dialog return value
var rej%;					'Flag to ignore spike from analysis

HideAll();					'Hide script window and toolbars
ToolbarVisible(1);	

data% := FrontView();					'If current view is a data file use it
if Viewkind(data%) <> 0 then			'Otherwise select data file
	data% := FileOpen("",0,2);			'Prompt user to select file		
endif;

if data% < 0 then						'If unable to open a file
	Message("No file opened\nScript closing");		'Warn user
	Halt;								'End script
endif;				
					
Window(0,0,100,100);			'Maximise the data view
XRange(MinTime(),MaxTime());	'Show whole time range
FrontView(data%);				'Make sure data file is current
DlgCreate("Spike analysis");	
DlgChan(1,"Select channel to analyse",1);		'Select channel to analyse
DlgCheck(2,"Check individual spikes");			'Option to check individual spikes
ok% := DlgShow(chan%,check%);			
if ok% = 0 then					'If user Cancels
	Halt;						'Halt script
endif;

res% := FileNew(1,0);			'Create a new text file for results
WindowTitle$("Results");		'..and title it
Window(0,70,100,100);			
Print("Frame\tSpike\tTime\tThreshold\tAmplitude\t10-90 Rise\tHalf width\tAHP\n");	'Print headers to the result
View(data%);
CursorDelete(-1);
HCursorDelete(-1);		'Delete existing horizontal cursors
HCursorNew(chan%);				'Set horizontal cursor on channel 
Interact("Set horizontal cursor to cut upstroke of all spikes and press OK",1023); 'Allow user to set level
CursorSet(7);																		'Set cursors to show detected positions
CursorLabel(4,1,"Peak");															'Label cursors
CursorLabel(4,2,"Threshold");
CursorLabel(4,3,"10% rise");
CursorLabel(4,4,"90% rise");
CursorLabel(4,5,"50% rise");
CursorLabel(4,6,"50% fall");
CursorLabel(4,7,"AHP");
CursorLabelPos(1,10);																'Set label positions to avoid overlapping
CursorLabelPos(2,75);
CursorLabelPos(3,70);
CursorLabelPos(4,20);
CursorLabelPos(5,30);
CursorLabelPos(6,40);
CursorLabelPos(7,80);

MinMax(chan%,MinTime(),MaxTime(),minData,maxData);			'Get full Y range of data in initial frame
pDAmp := (maxData-minData)/50.0;							'Calculate default value for use in peak detection

for frameCounter% := 1 to FrameCount() do			'Loop through each frame
	count% := 0;											'Reset spike counter
	Frame(frameCounter%);								'Go to frame	
	DrawAll();
	current := MinTime();								'Set variable to beginning of frame
	test := MinTime();												'Initial check to see if any spikes present
	if ChanSearch(chan%,7,test,MaxTime(),HCursor(1)) = 0 then		'Print header based on presence of spikes in frame
		View(res%).Print("%d",frameCounter%);
	else
		View(res%).Print("%d\n\n",frameCounter%);
	endif;
	repeat													'Loop through each crossing	
		gotSpikeFlag% := ChanSearch(chan%,7,current,MaxTime(),HCursor(1));
		if gotSpikeFlag% = 0 then							'If spike detected
			ArrConst(error%[],0);							'Reset error count
			err% := 0;										'Reset counter
			rej% := 0;										'Reset rejection flag
			pTime := current;								'Set peak time variable to start time
			got% := ChanSearch(chan%,4,pTime,MaxTime(),0,pDAmp);	'Find spike peak
			if got% < 0 then error%[0] := 1;endif;			'Set value of error in peak find
			Cursor(1,pTime);								'Set cursor at peak position
			rTime := pTime;									'Set max slope time variable to peak time
			ChanSearch(chan%,9,rTime,rTime-0.005);			'Find maximum slope
			sTime := rTime;									'Set spike start variable to slope time
			got% := ChanSearch(chan%,18,sTime,sTime-0.005,0,20);		'Find start of spike
			if got% < 0 then error%[1] := 1;endif;
			Cursor(2,sTime);								'Set cursor at spike start
			spikeTimes[count%] := sTime;					'Put time into array
			pAmp := ChanValue(chan%,pTime);					'Get amplitudes at peak and start of spike
			sAmp := ChanValue(chan%,sTime);
			amp := pAmp - sAmp;								'Get spike amplitude relative to start
			riseAmp10 := sAmp + (amp * 0.1);				'Calculate levels for 10%, 50% and 90% rise
			riseAmp90 := sAmp + (amp * 0.9);
			riseAmp50 := sAmp + (amp * 0.5);
			riseTime10 := sTime;									
			got% := ChanSearch(chan%,7,riseTime10,pTime,riseAmp10);		'Find time at 10% rise
			if got% < 0 then error%[2] := 1;endif;
			Cursor(3,riseTime10);
			riseTime90 := sTime;
			got% := ChanSearch(chan%,7,riseTime90,pTime,riseAmp90);		'Find time at 90% rise
			if got% < 0 then error%[3] := 1;endif;
			Cursor(4,riseTime90);
			riseTime50 := sTime;
			got% := ChanSearch(chan%,7,riseTime50,pTime,riseAmp50);		'Find time at 50% rise
			if got% < 0 then error%[4] := 1;endif;
			Cursor(5,riseTime50);
			fallTime50 := pTime;
			got% := ChanSearch(chan%,8,fallTime50,pTime + 0.005,riseAmp50);	'Find time at 50% fall
			if got% < 0 then error%[5] := 1;endif;
			Cursor(6,fallTime50);
			hWidth := fallTime50 - riseTime50;						'Half width time
			AHPTime := fallTime50;							
			got% := ChanSearch(chan%,5,AHPTime,MaxTime(),0,pDAmp/5);		''Find trough following spike
			if got% < 0 then error%[6] := 1;endif;
			AHPAmp := ChanValue(chan%,AHPTime);
			Cursor(7,AHPTime);
			count% += 1;									'Add 1 to spike count
			numErrors% := ArrSum(error%[]);
		 	if  numErrors% > 0 then
				check% := 1;
			endif;
			if check% > 0 then														
				XRange(current - 0.005,AHPTime + 0.005);
				if numErrors% > 0 then											'Routine to detect search errors and allow user to manually set positions
					all% := 1;													'Set variable to ensure we do not change display if spike rejected
					DlgCreate("SEARCH ERROR!!!",60,5);								'Set up dialog to be displayed if detection of any points failed
					DlgText("Failure in detection of following points",15,2);
					DlgText("May cause additional undetected errors",14,3);
					for i% := 1 to 7 do											'Display failed feature
						if error%[i%-1] > 0 then
							err% += 1;
							DlgText(pos$[i%-1],2,err%+4);
						endif;
					next;
					DlgText("Manually position cursors and continue, or reject spike from analysis",2,err%+6);
					DlgAllow(1023);
					DlgButton(1,"Continue");
					DlgButton(0,"Reject");
					ok% := DlgShow();
					if ok% = 0 then
						rej% := 1;
					endif;
				else
					all% := Interact("Check and adjust positions and proceed to next spike or run through all spikes",1023,0,"Next|0x0d","Run");
				endif;
				pAmp := ChanValue(chan%,Cursor(1));					'Recalculate values based on current cursor positions
				sTime := Cursor(2);
				sAmp := ChanValue(chan%,sTime);
				amp := pAmp - sAmp;
				riseTime10 := Cursor(3);
				riseTime90 := Cursor(4);
				riseTime50 := Cursor(5);
				fallTime50 := Cursor(6);
				AHPAmp := ChanValue(chan%,Cursor(7));
				if all% = 2 then
					View(data%).XRange(MinTime(),MaxTime());
					DrawAll();
					check% := 0;
				endif;
			endif;
			if rej% = 0 then
				View(res%).Print("\t%d\t%g\t%g\t%g\t%g\t%g\t%g\n",count%,sTime*1000.0,sAmp,amp,(riseTime90-riseTime10)*1000000.0,hWidth*1000000.0,AHPAmp);	'Print detailed results to text window
			else
				count% -= 1;
			endif;
		endif;													
	until gotSpikeFlag% < 0;							'Stop when no spike found		

	if count% > 0 then
		View(res%).Print("\n");
	endif;
next;
View(data%);
Window(0,0,100,70);
FrontView(res%);
RestoreAll();

proc HideAll()
var i%;
gFloat%[0] := App(-2);        'number of windows
for i% := 1 to gFloat%[0] do  'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do  'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end

Did you know…? 
1902 conditioner control panel in Spike2 and Signal has now been revised to fully support 
the new 1902 mk IV units which uses digital filtering, rather than the pre-set hardware 
filters used in previous models. The user can select between 2 or 3-pole Bessel and 
Butterworth filter types and the cut-off frequencies can be set either from the drop down 
lists, or entered as values.  

 
1902 control panel for digital filters 

 

Recent questions  
Q. Is it possible to interface with visual stimulus generation systems, other 

application software and external devices with Signal, Spike2 and a 1401?  

A. There are a number of ways that Spike2 and Signal can interact with external 
devices and other software applications to allow complete experiment control. A 
1401 can be used to generate digital outputs, bit patterns and waveform outputs 
to trigger and control stimulus generators while also reacting to incoming digital 
and waveform data in real-time. Both Spike2 and Signal can also access and 
output serial-line data to control and respond to more specialist external 
equipment such as audiometers or TMS units. Script users are able to run, close 
and check the status of other software applications directly from Spike2. We can 
also supply a ‘listener’ specification that can be used to define an interface 
whereby an external program can track and synchronise with Spike2 data 
sampling. This is the same specification that was used to create our Spike2 
Video application and allows Windows programmers to write their own external 
applications to synchronise with Spike2. See the on-line help for detailed 
descriptions of our output and control capabilities. 

